# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Daniel Wallace <danielwallace at gtmanfred dot com>

pkgname=python-octaviaclient
pkgver=3.1.0
_commit=03083c3dabdd4dc0c945a8cfcca0b9ba8852063e
pkgrel=2
pkgdesc="Octavia client for OpenStack Load Balancing"
arch=('any')
url="https://docs.openstack.org/python-octaviaclient/latest/"
license=('Apache')
depends=('python-cliff' 'python-keystoneauth1' 'python-neutronclient' 'python-openstackclient'
         'python-osc-lib' 'python-oslo-serialization' 'python-oslo-utils' 'python-pbr'
         'python-requests' 'python-munch')
makedepends=('git')
checkdepends=('python-hacking' 'python-oslotest' 'python-requests-mock')
source=("git+https://github.com/openstack/python-octaviaclient.git#commit=$_commit")
sha512sums=('SKIP')

build() {
  cd python-octaviaclient
  python setup.py build
}

check() {
  cd python-octaviaclient
  stestr run
}

package() {
  cd python-octaviaclient
  python setup.py install --root="$pkgdir" --optimize=1
}
